

Περισσότερη Prolog



Tamara de Lempicka

Κωστής Σαγώνας <kostis@cs.ntua.gr>

Περιεχόμενα

- Ενοποίηση
- Μοντέλα εκτέλεσης της Prolog
 - Το διαδικαστικό μοντέλο
 - Το μοντέλο υλοποίησης
 - Το αφηρημένο μοντέλο
- Περισσότερη Prolog
 - Κατηγορήματα εισόδου και εξόδου
 - Κατηγορήματα διαχείρισης της βάσης της Prolog
 - Αριθμητικοί υπολογισμοί και συγκρίσεις στην Prolog
- Παραδείγματα προγραμματισμού σε Prolog
- Αποχαιρετισμός στην Prolog

Περισσότερη Prolog

2

Αντικαταστάσεις (Substitutions)

- Μια *αντικατάσταση* είναι μια συνάρτηση που απεικονίζει μεταβλητές σε όρους:

$$\sigma = \{X \rightarrow a, Y \rightarrow f(a, b)\}$$

- Η αντικατάσταση σ αντιστοιχεί την μεταβλητή X στο a και τη Y στο $f(a, b)$
- Το αποτέλεσμα της εφαρμογής μιας αντικατάστασης σε έναν όρο δημιουργεί ένα *στιγμιότυπο* του όρου
- Για παράδειγμα $\sigma(g(X, Y)) = g(a, f(a, b))$
- Ο όρος $g(a, f(a, b))$ είναι στιγμιότυπο του $g(X, Y)$

Περισσότερη Prolog

3

Ενοποίηση

- Δύο όροι της Prolog t_1 και t_2 *ενοποιούνται* εάν υπάρχει κάποια αντικατάσταση σ (ο *ενοποιητής* τους) που κάνει τους δύο όρους ακριβώς τους ίδιους: $\sigma(t_1) = \sigma(t_2)$
 - Οι όροι a και b δεν ενοποιούνται
 - Οι όροι $f(X, b)$ και $f(a, Y)$ ενοποιούνται: ένας ενοποιητής είναι ο $\{X \rightarrow a, Y \rightarrow b\}$
 - Οι όροι $f(X, b)$ και $g(Y, b)$ δεν ενοποιούνται
 - Οι όροι $a(X, X, b)$ και $a(b, Y, Y)$ ενοποιούνται: ένας ενοποιητής είναι ο $\{X \rightarrow b, Y \rightarrow b\}$
 - Οι όροι $a(X, X, b)$ και $a(c, Y, Y)$ δεν ενοποιούνται
 - Οι όροι $a(X, f)$ και $a(Y, f)$ ενοποιούνται: ένας ενοποιητής είναι ο $\{X \rightarrow 42, Y \rightarrow 42\}$

Περισσότερη Prolog

4

Πολλαπλοί ενοποιητές

- **parent(X, Y)** και **parent(fred, Z)**:
 - Ένας ενοποιητής είναι ο $\sigma_1 = \{X \rightarrow fred, Y \rightarrow Z\}$
 - Άλλος ένας είναι ο $\sigma_2 = \{X \rightarrow fred, Y \rightarrow mary, Z \rightarrow mary\}$
 - Άλλος ένας είναι ο $\sigma_3 = \{X \rightarrow fred, Y \rightarrow foo(42), Z \rightarrow foo(42)\}$
- Η Prolog επιλέγει ενοποιητές όπως ο σ_1 , οι οποίοι καθορίζουν ακριβώς τις αντικαταστάσεις που είναι αναγκαίες για την ενοποίηση των δύο όρων
- Με άλλα λόγια, επιλέγει τον πιο γενικό ενοποιητή των δύο όρων (MGU – Most General Unifier)

Περισσότερη Prolog

5

Ο Πιο Γενικός Ενοποιητής (ΠΓΕ)

- Ο όρος x_1 είναι *πιο γενικός* από τον όρο x_2 εάν ο x_2 είναι στιγμιότυπο του x_1 , αλλά ο x_1 δεν είναι στιγμιότυπο του x_2
 - Παράδειγμα: ο όρος **parent(fred, Y)** είναι πιο γενικός από τον όρο **parent(fred, mary)**
- Ένας ενοποιητής σ_1 δύο όρων t_1 και t_2 είναι ο *πιο γενικός ενοποιητής* εάν δεν υπάρχει άλλος ενοποιητής σ_2 τέτοιος ώστε ο όρος $\sigma_2(t_i)$ να είναι πιο γενικός από τον όρο $\sigma_1(t_i)$
- Μπορεί να αποδειχθεί ότι ο πιο γενικός ενοποιητής είναι μοναδικός (με εξαίρεση τα ονόματα των μεταβλητών)

Περισσότερη Prolog

6

Η Prolog χρησιμοποιεί ενοποίηση για τα πάντα

- Πέρασμα παραμέτρων
 - `reverse([1,2,3],X)`
- Δέσιμο μεταβλητών
 - `X = 0`
- Κατασκευή δεδομένων
 - `X = .(1,[2,3])`
- Επιλογή δεδομένων
 - `[1,2,3] = .(X,Y)`

Περισσότερη Prolog

7

Έλεγχος Εμφάνισης (Occurs Check)

- Μια μεταβλητή x και ένας όρος t ενοποιούνται με την αντικατάσταση $\{x \rightarrow t\}$:
 - x και b ενοποιούνται: ο ΠΓΕ είναι $\{x \rightarrow b\}$
 - x και $f(a, g(b))$ ενοποιούνται: ο ΠΓΕ είναι $\{x \rightarrow f(a, g(b))\}$
 - x και $f(a, Y)$ ενοποιούνται: ο ΠΓΕ είναι $\{x \rightarrow f(a, Y)\}$
- *Εκτός εάν η μεταβλητή x περιλαμβάνεται στον όρο t :*
 - Οι όροι x και $f(a, g(x))$ δεν ενοποιούνται: η αντικατάσταση $\{x \rightarrow f(a, g(x))\}$ δεν είναι ενοποιητής
- Με άλλα λόγια, τουλάχιστον στη θεωρία, η ενοποίηση πρέπει να είναι καλά θεμελιωμένη (well-founded)

Περισσότερη Prolog

8

Παράδειγμα Occurs Check

- Η default ενοποίηση της Prolog δεν περιλαμβάνει έλεγχο εμφάνισης

```
append([], L, L).
append([H|L1], L2, [H|L3]) :-
    append(L1, L2, L3).
```

```
?- append([], X, [a | X]).
```

```
X = [a, a, a, a, a, a, a, a, a | ...] ;
```

```
No
```

- Αλλά το ISO Prolog standard περιλαμβάνει ένα κατηγορήμα με όνομα `unify_with_occurs_check/2`

Περισσότερη Prolog

9

Μοντέλα Εκτέλεσης της Prolog

Περισσότερη Prolog

10

Το Διαδικαστικό Μοντέλο Εκτέλεσης της Prolog

- Κάθε κατηγορήμα είναι μια διαδικασία για την απόδειξη στόχων
 - `p :- q, r.`
 - Για την απόδειξη του στόχου p , πρώτα ενοποίησε το στόχο με την κεφαλή του κανόνα p , μετά απέδειξε το q , και μετά απέδειξε το r
 - `s.`
 - Για την απόδειξη του στόχου s , ενοποίησε το στόχο με το s
- Η κάθε πρόταση (γεγονός ή κανόνας) αποτελεί ένα διαφορετικό τρόπο απόδειξης του στόχου
- Η απόδειξη μπορεί να περιλαμβάνει κλήσεις σε άλλα κατηγορήματα-διαδικασίες

Περισσότερη Prolog

11

Παραδείγματα

```
p :- q, r.          boolean p() {return q() && r();}
q :- s.            boolean q() {return s();}
r :- s.            boolean r() {return s();}
s.                 boolean s() {return true;}
```

```
p :- p.            boolean p() {return p();}
```

Περισσότερη Prolog

12

Οπισθοδρόμηση (Backtracking)

- Μια περιπλοκή: οπισθοδρόμηση
- Η Prolog κρατάει μια λίστα με όλους τους δυνατούς τρόπους με τους οποίους μπορεί να ικανοποιηθεί κάποιος στόχος και τους δοκιμάζει με τη σειρά μέχρι να ικανοποιηθεί πλήρως το στόχο ή μέχρι να εξαντλήσει όλες τις δυνατότητες ικανοποίησής του
- Έστω ο στόχος p στο παρακάτω πρόγραμμα: ο στόχος ικανοποιείται αλλά μόνο με χρήση οπισθοδρόμησης

```

1. p :- q, r.
2. q :- s.
3. q.
4. r.
5. s :- 0 = 1.
    
```

Περισσότερη Prolog


13

Αντικατάσταση

- Άλλη μια περιπλοκή: αντικατάσταση μεταβλητών
- Μια κρυμμένη ροή πληροφορίας

Η αντικατάσταση $\sigma_1 = \text{MGU}(p(f(Y)), t)$ εφαρμόζεται σε όλες τις μεθεπόμενες συνθήκες προς ικανοποίηση της πρότασης

Η αντικατάσταση σ_3 προκύπτει από την απόδειξη του όρου $\sigma_2(\sigma_1(x(Y)))$
 Η σύνθεση των αντικαταστάσεων επιστρέφεται στον καλούντα
 Ο όρος που αποδείχθηκε είναι ο $\sigma_3(\sigma_2(\sigma_1(t)))$



Περισσότερη Prolog

14

Το Μοντέλο Υλοποίησης: Διάλυση (Resolution)

- Το βασικό βήμα συμπερασμού
- Κάθε κανόνας αναπαριστάται με μια λίστα από όρους (κάθε γεγονός αναπαριστάται από λίστα ενός στοιχείου)
- Κάθε βήμα διάλυσης χρησιμοποιεί μια από αυτές τις λίστες, μια φορά, για να επιτύχει κάποια πρόοδο στην απόδειξη μιας λίστας από όρους που είναι στόχοι προς απόδειξη για την απάντηση κάποιας ερώτησης

```

function resolution(clause, goals):
    let sub = the MGU of head(clause) and head(goals)
    return sub(body(clause) concatenated with tail(goals))
    
```

Περισσότερη Prolog

15

Παράδειγμα Διάλυσης

Για την παρακάτω λίστα όρων προς απόδειξη:
 $[p(x), s(x)]$
 και τον κανόνα:
 $p(f(y)) :- q(y), r(y).$
 Ο ΠΓΕ είναι ο $\{x \rightarrow f(y)\}$, και στο επόμενο βήμα προκύπτει η λίστα αποτελούμενη από τους όρους:
 $\text{resolution}([p(f(y)), q(y), r(y)], [p(x), s(x)]) = [q(y), r(y), s(f(y))]$

```

function resolution(clause, goals):
    let sub = the MGU of head(clause) and head(goals)
    return sub(body(clause) concatenated with tail(goals))
    
```

Περισσότερη Prolog

16

Ένας διερμηνέας της Prolog

```

function solve(goals)
    if goals is empty then succeed()
    else for each clause c in the program, in order
        if head(c) does not unify with head(goals) then do nothing
        else solve(resolution(c, goals))
    
```

Περισσότερη Prolog

17

Παράδειγμα

Μια μερική επίλυση του $p(x)$:

<pre> 1. p(f(Y)) :- q(Y), r(Y) . 2. q(g(Z)) . 3. q(h(Z)) . 4. r(h(a)) . </pre>	<pre> solve([p(x)]) 1. solve([q(Y), r(Y)]) ... 2. nothing 3. nothing 4. nothing </pre>
---	---

- Η συνάρτηση `solve` δοκιμάζει τις τέσσερις προτάσεις τη μία μετά την άλλη
 - Η πρώτη πρόταση δουλεύει, και η συνάρτηση `solve` καλεί τον εαυτό της αναδρομικά με το αποτέλεσμα της διάλυσης
 - Οι άλλες τρεις προτάσεις δε δουλεύουν: οι κεφαλές τους δεν ενοποιούνται με τον όρο της λίστας

Περισσότερη Prolog

18

Παραδείγματος συνέχεια

```
1. p(f(Y)) :-
   q(Y), r(Y).
2. q(g(Z)).
3. q(h(Z)).
4. r(h(a)).
```

Μια μερική επίλυση του $p(X)$, επεκταμένη:

```
solve([p(X)])
1. solve([q(Y), r(Y)])
   1. nothing
   2. solve([r(g(Z))])
      ...
   3. solve([r(h(Z))])
      ...
   4. nothing
2. nothing
3. nothing
4. nothing
```

Περισσότερη Prolog

19

Το παράδειγμα ολοκληρωμένο

```
1. p(f(Y)) :-
   q(Y), r(Y).
2. q(g(Z)).
3. q(h(Z)).
4. r(h(a)).
```

Η ολική επίλυση της ερώτησης $p(X)$:

```
solve([p(X)])
1. solve([q(Y), r(Y)])
   1. nothing
   2. solve([r(g(Z))])
      1. nothing
      2. nothing
      3. nothing
      4. nothing
   3. solve([r(h(Z))])
      1. nothing
      2. nothing
      3. nothing
      4. solve([]) -success!
   4. nothing
2. nothing
3. nothing
4. nothing
```

Περισσότερη Prolog

20

Συλλογή των αντικαταστάσεων

```
function resolution(clause, goals, query):
  let sub = the MGU of head(clause) and head(goals)
  return (sub(tail(clause) concatenated with tail(goals)), sub(query))

function solve(goals, query)
  if goals is empty then succeed(query)
  else for each clause c in the program, in order
       if head(c) does not unify with head(goals) then do nothing
       else solve(resolution(c, goals, query))
```

- Τροποποιημένη συνάρτηση που δέχεται ως όρισμα την αρχική ερώτηση και εφαρμόζει όλες τις αντικαταστάσεις πάνω του
- Στο τέλος της διάλυσης, το αποδειχθέν στιγμιότυπο περνιέται ως όρισμα στη συνάρτηση **succeed**

Περισσότερη Prolog

21

```
1. p(f(Y)) :-
   q(Y), r(Y).
2. q(g(Z)).
3. q(h(Z)).
4. r(h(a)).
```

Η ολική επίλυση της ερώτησης $p(X)$:

```
solve([p(X)], p(X))
1. solve([q(Y), r(Y)], p(f(Y)))
   1. nothing
   2. solve([r(g(Z))], p(f(g(Z))))
      1. nothing
      2. nothing
      3. nothing
      4. nothing
   3. solve([r(h(Z))], p(f(h(Z))))
      1. nothing
      2. nothing
      3. nothing
      4. solve([], p(f(h(a))))
   4. nothing
2. nothing
3. nothing
4. nothing
```

Περισσότερη Prolog

22

Διερμηνείς της Prolog

- Ο διερμηνέας που μόλις είδαμε είναι λειτουργικός αλλά δεν χρησιμοποιείται από τις υλοποιήσεις της Prolog
- Όλες οι υλοποιήσεις της Prolog πρέπει μεν να λειτουργήσουν με τον τρόπο που μόλις περιγράψαμε αλλά συνήθως χρησιμοποιούν εντελώς διαφορετικές τεχνικές υλοποίησης: μεταφράζουν τον κώδικα σε γλώσσα κάποιας αφηρημένης μηχανής
- Η πιο συνηθισμένη τέτοια μηχανή είναι η Αφηρημένη Μηχανή του Warren (Warren Abstract Machine)

Περισσότερη Prolog

23


Το Αφηρημένο Μοντέλο: Δένδρα Απόδειξης

- Θέλουμε να αναπαραστήσουμε με κάποιο τρόπο τη σειρά των λειτουργιών της εκτέλεσης, όμως χωρίς να περιορίζεται η τεχνική της υλοποίησης
- Τα δένδρα απόδειξης αποτελούν έναν τέτοιο formalισμό:
 - Η ρίζα είναι η αρχική ερώτηση
 - Οι κόμβοι του δένδρου είναι λίστες από όρους προς απόδειξη, και κάθε κόμβος έχει ένα παιδί για κάθε πρόταση του προγράμματος

Περισσότερη Prolog

24

Παράδειγμα



Περισσότερη Prolog

25


Απλοποίηση των δένδρων απόδειξης

- Τα παιδιά κάθε κόμβου αντιστοιχούν στις προτάσεις του προγράμματος
- Και η σειρά τους είναι ίδια με τη σειρά εμφάνισής τους στο πρόγραμμα
- Μπορούμε να απλοποιήσουμε το δένδρο με το να απαλείψουμε όλους τους *nothing* κόμβους
- Οι κόμβοι αυτοί αντιστοιχούν σε προτάσεις οι οποίες δεν ενοποιούνται με το πρώτο στοιχείο της λίστας του κόμβου πατέρα

Περισσότερη Prolog

26

Παράδειγμα απλοποιημένου δένδρου απόδειξης



Περισσότερη Prolog

27

Σημασιολογία της Prolog

- Δοθέντος ενός προγράμματος και μιας ερώτησης, ένα σύστημα Prolog πρέπει να λειτουργήσει με τρόπο που καθορίζεται από μια πρώτα κατά βάθος, αριστερά προς τα δεξιά διάσχιση (depth-first, left-to-right traversal) του δένδρου απόδειξης
- Ένας τρόπος υλοποίησης είναι μέσω κάποιου διερμηνέα όπως αυτός που ορίζεται από τη συνάρτηση `solve`
- Στην πράξη, συνήθως χρησιμοποιούνται άλλοι, πιο αποδοτικοί, τρόποι υλοποίησης

Περισσότερη Prolog

28

Κατηγορήματα Εισόδου και Εξόδου

- Η είσοδος και η έξοδος όρων γίνεται σχετικά απλά

```
?- write('Hello world').
Hello world

Yes
?- read(X).
|: hello(world).

X = hello(world) ;

No
```

- Οποιοσδήποτε όρος της Prolog μπορεί να διαβαστεί
- Υπάρχει επίσης το κατηγορήμα `n1/0` το οποίο είναι ισοδύναμο με την κλήση `write('\n')`

Περισσότερη Prolog

29

Το κατηγορήμα `assert/1`

- Εισάγει ένα γεγονός ή κανόνα στην εσωτερική βάση δεδομένων της Prolog (στο τέλος της βάσης)

```
?- dynamic parent/2.

Yes
?- parent(X,Y).

No
?- assert(parent(joe,mary)).

Yes
?- parent(X,Y).

X = joe
Y = mary ;

No
```

Περισσότερη Prolog

30

Το κατηγορήμα `retract/1`

- Απομακρύνει την πρώτη πρόταση (κανόνα ή γεγονός) που ενοποιείται με το όρισμά του από τη βάση των δεδομένων

```
?- parent(joe,mary) .
Yes
?- retract(parent(joe,mary)) .
Yes
?- parent(joe,mary) .
No
```

- Υπάρχει επίσης και το κατηγορήμα `retractall/1` το οποίο απομακρύνει όλες τις προτάσεις που ενοποιούνται με το όρισμά του

Περισσότερη Prolog

31

Μη αποτιμημένοι όροι

- Οι τελεστές της Prolog επιτρέπουν πιο συμπυκνωμένη γραφή των όρων, αλλά οι όροι δεν αποτιμούνται
- Όλα οι παρακάτω γραφές είναι ο ίδιος όρος Prolog:

```
1+ *(2,3)
+(1,2*3)
(1+(2*3))
1+2*3
```

- Ο παραπάνω όρος *δεν* ενοποιείται με τον όρο 7

Περισσότερη Prolog

32

Αποτίμηση αριθμητικών εκφράσεων

```
?- X is 1+2*3.
X = 7 ;
No
```

- Το προκαθορισμένο κατηγορήμα `is/2` μπορεί να χρησιμοποιηθεί για την αποτίμηση ενός όρου που είναι μια αριθμητική έκφραση
- `is (X, Y)` αποτιμά τον όρο `Y` και ενοποιεί τον `X` με το αποτέλεσμα της αποτίμησης
- Συνήθως χρησιμοποιείται ως διαδικός τελεστής

Περισσότερη Prolog

33

Η αποτίμηση προϋποθέτει τιμές...

```
?- Y = X+2, X = 1.
Y = 1+2
X = 1 ;
No
?- Y is X+2, X = 1.
ERROR: Arguments are not sufficiently instantiated
?- X = 1, Y is X+2.
X = 1
Y = 3 ;
No
```

Περισσότερη Prolog

34

Αριθμητικές εκφράσεις και συναρτήσεις

- Σε έναν όρο `X is Y`, οι υπό-όροι του `Y` πρέπει να είναι *αποτιμώμενες συναρτήσεις*
- Αυτές περιλαμβάνουν τους προκαθορισμένους αριθμητικούς τελεστές `+`, `-`, `*` και `/`
- Όπως και προκαθορισμένες αριθμητικές συναρτήσεις, π.χ. `abs(X)`, `sqrt(X)`, `floor(X)`, ...

Περισσότερη Prolog

35

Πραγματικές και ακέραιες τιμές

```
?- X is 1/2.
X = 0.5 ;
No
?- X is 1.0/2.0.
X = 0.5 ;
No
?- X is 2 // 1.
X = 2 ;
No
?- X is 2.0/1.0.
X = 2.0 ;
No
```

Δύο αριθμητικοί τύποι: ακέραιοι και πραγματικοί.

Οι περισσότερες αριθμητικές συναρτήσεις είναι υπερφορτωμένες για όλους τους συνδυασμούς ορισμάτων.

Η Prolog έχει δυναμικούς τύπους – οι τύποι χρησιμοποιούνται κατά το χρόνο εκτέλεσης για την επίλυση της υπερφόρτωσης.

Παρατηρείστε όμως ότι ο στόχος `2 = 2.0` αποτυγχάνει.

Περισσότερη Prolog

36

Αριθμητικές συγκρίσεις

- Τελεστές αριθμητικής σύγκρισης:
<, >, =<, >=, :=, :=
- Σε μια αριθμητική σύγκριση, η Prolog αποτιμά και τα δύο ορίσματα του τελεστή και συγκρίνει τις τιμές που προκύπτουν
- Άρα και τα δύο ορίσματα πρέπει να έχουν τιμές για όλες τις μεταβλητές τους

Περισσότερη Prolog

37

Αριθμητικές συγκρίσεις

```
?- 1+2 < 1*2.
No
?- 1 < 2.0.
Yes
?- 1+2 >= 1+3.
No
?- X is 1-3, Y is 0-2, X := Y.
X = -2
Y = -2 ;
No
```

Περισσότερη Prolog

38

Ισότητες στην Prolog

- Μέχρι στιγμής έχουμε χρησιμοποιήσει τρεις διαφορετικούς τελεστές ισότητας:
 X is Y αποτιμά τον όρο Y και ενοποιεί το αποτέλεσμα με το X
 π.χ. 3 is $1+2$ επιτυγχάνει, αλλά $1+2$ is 3 αποτυγχάνει
 $X = Y$ ενοποιεί τους όρους X και Y , αλλά δεν τους αποτιμά
 π.χ. τόσο ο στόχος $3 = 1+2$ όσο και ο $1+2 = 3$ αποτυγχάνουν
 $X := Y$ αποτιμά τους δύο όρους και τους συγκρίνει αριθμητικά
 π.χ. τόσο ο στόχος $3 := 1+2$ και ο $1+2 := 3$ επιτυγχάνουν

Περισσότερη Prolog

39

Παραδείγματα

```
mylength1([],0).
mylength1([_|T],Len) :-
  mylength1(T,LenT),
  Len = LenT + 1.
```

```
mylength2([],0).
mylength2([_|T],Len) :-
  mylength2(T,LenT),
  Len is LenT + 1.
```

```
?- mylength1([a,b,c],X).
X = 0+1+1+1 ;
No
```

```
?- mylength2([a,b,c],X).
X = 3 ;
No
?- mylength2(X,3).
X = [_ , _ , _] ;
No
```

Περισσότερη Prolog

40

Παράδειγμα: sum

```
sum([],0).
sum([Head|Tail],Sum) :-
  sum(Tail,TailSum),
  Sum is Head + TailSum.
```

```
?- sum([1, 2, 3], X).
X = 6 ;
No
?- sum([1, 2.5, 3], X).
X = 6.5 ;
No
```

Περισσότερη Prolog

41

Παράδειγμα: gcd

```
gcd(X,Y,GCD) :- ← Προσοχή: όχι απλά
  X := Y,
  GCD is X.
gcd(X,Y,GCD) :-
  X < Y,
  NewY is Y - X,
  gcd(X,NewY,GCD).
gcd(X,Y,GCD) :-
  X > Y,
  NewX is X - Y,
  gcd(NewX,Y,GCD).
gcd(X,X,X)
```

Περισσότερη Prolog

42

Παράδειγμα: χρήση του κατηγορήματος gcd

```
?- gcd(5,5,X).
X = 5 ;
No
?- gcd(12,21,X).
X = 3 ;
No
?- gcd(91,105,X).
X = 7 ;
No
?- gcd(91,21*5,7).
Yes
?- gcd(91,X,7).
ERROR: Arguments are not sufficiently instantiated
```

Παράδειγμα: factorial

```
factorial(X,Fact) :-
  X =:= 1, Fact is 1.
factorial(X,Fact) :-
  X > 1,
  NewX is X - 1,
  factorial(NewX,NF),
  Fact is X * NF.
```

```
?- factorial(5,F).
F = 120 ;
No
?- factorial(10,F).
F = 3628800 ;
No
?- factorial(-2,F).
No
```

Διάζευξη και if-then-else στην Prolog

- Ο δυαδικός τελεστής ;/2 υλοποιεί τη διάζευξη (or)
- Ο ορισμός του είναι:

```
';' (X, _) :- X.
';' (_, Y) :- Y.
```

- Το if-then-else γράφεται με χρήση διάζευξης και του δυαδικού τελεστή ->/2
- Ο ορισμός του είναι:

```
(Cond -> Then ; _) :- Cond, Then.
(Cond -> _ ; Else) :- \+ Cond, Else.
```

Παράδειγμα χρήσης διάζευξης και if-then-else

```
factorial(X,Fact) :-
  X =:= 1, Fact is 1.
factorial(X,Fact) :-
  X > 1,
  NewX is X - 1,
  factorial(NewX,NF),
  Fact is X * NF.
```



```
factorial(X,Fact) :-
  ( X =:= 1, Fact is 1
  ; X > 1,
    NewX is X - 1,
    factorial(NewX,NF),
    Fact is X * NF,
  ).
```



```
factorial(X,Fact) :-
  ( X =:= 1 -> Fact is 1
  ; X > 1,
    NewX is X - 1,
    factorial(NewX,NF),
    Fact is X * NF,
  ).
```

Παράδειγμα χρήσης if-then-else


```
gcd(X,Y,GCD) :-
  X =:= Y,
  GCD is X.
gcd(X,Y,GCD) :-
  X < Y,
  NewY is Y - X,
  gcd(X,NewY,GCD).
gcd(X,Y,GCD) :-
  X > Y,
  NewX is X - Y,
  gcd(NewX,Y,GCD).
```



```
gcd(X,Y,GCD) :-
  ( X =:= Y ->
    GCD is X
  ; X < Y ->
    NewY is Y - X,
    gcd(X,NewY,GCD)
  ; X > Y ->
    NewX is X - Y,
    gcd(NewX,Y,GCD)
  ).
```

Από τον παραπάνω κώδικα μπορούμε να παραλείψουμε τον έλεγχο της συνθήκης $X > Y$ (και φυσικά το $->$)

Δύο Παραδείγματα Προγραμμάτων



Προβλήματα Αναζήτησης Λύσης

- Η Prolog δε σχεδιάστηκε για προγραμματισμό αριθμητικών εφαρμογών (προφανώς)
- Τα προβλήματα στα οποία η Prolog δείχνει τις ικανότητές της είναι προβλήματα που χρησιμοποιούν αναζήτηση σε ένα χώρο λύσεων και όπου
 - Προσδιορίζουμε έναν ορισμό της λύσης με χρήση λογικής και
 - Αφήνουμε την Prolog να βρει αυτή τη λύση
- Θα εξετάσουμε τις λύσεις δύο προβλημάτων:
 - Το πρόβλημα του σακιδίου
 - Το πρόβλημα των οκτώ βασιλισσών

Περισσότερη Prolog

49

Το Πρόβλημα του Σακιδίου (Knapsack Problem)

- Ετοιμάζουμε το σακίδιό μας για μια εκδρομή
- Στα ντουλάπια υπάρχουν τα εξής τρόφιμα:

Item	Weight in kilograms	Calories
bread	4	9200
pasta	2	4600
peanut butter	1	6700
baby food	3	6900

- Το σακίδιο χωράει πράγματα συνολικού βάρους 4 kg.
- Ποια επιλογή πραγμάτων βάρους ≤ 4 kg. μεγιστοποιεί το ποσό των θερμίδων;

Περισσότερη Prolog

50

Οι άπληστες μέθοδοι δε δουλεύουν

Item	Weight in kilograms	Calories
bread	4	9200
pasta	2	4600
peanut butter	1	6700
baby food	3	6900

- Πρώτα τις περισσότερες θερμίδες: bread = 9200
- Πρώτα τα πιο ελαφριά: peanut butter + pasta = 11300
- (Η βέλτιστη επιλογή: peanut butter + baby food = 13600)

Περισσότερη Prolog

51

Αναζήτηση

- Δεν υπάρχει γνωστός αλγόριθμος για το συγκεκριμένο πρόβλημα που
 - Πάντα δίνει τη βέλτιστη λύση για το πρόβλημα, και
 - Χρειάζεται λιγότερο από εκθετικό χρόνο για να τη βρει
- Κατά συνέπεια, η εξαντλητική αναζήτηση δεν είναι κάτι για το οποίο θα πρέπει να ντρεπόμαστε για τη χρήση του
- Το αντίθετο μάλιστα, επειδή η αναζήτηση είναι κάτι που η Prolog κάνει καλά

Περισσότερη Prolog

52

Αναπαράσταση

- Θα αναπαραστήσουμε κάθε είδος φαγητού με τον όρο `food(N, W, C)`
- Το ντουλάπι του παραδείγματός μας αναπαρίσταται ως:


```
[ food(bread, 4, 9200),
  food(pasta, 2, 4500),
  food(peanutButter, 1, 6700),
  food(babyFood, 3, 6900) ]
```
- Θα χρησιμοποιήσουμε την ίδια αναπαράσταση για τα περιεχόμενα του σακιδίου

Περισσότερη Prolog

53

Ακολουθίες και υπακολουθίες

```
/*
  subseq(X,Y) succeeds when list X is the same as
  list Y, but with zero or more elements omitted.
  This can be used with any pattern of instantiations.
*/
subseq([], []).
subseq([Item | RestX], [Item | RestY]) :-
  subseq(RestX, RestY).
subseq(X, [_ | RestY]) :-
  subseq(X, RestY).
```

- Μια υπακολουθία μιας λίστας είναι ένα αντίγραφο της λίστας όπου κάποια στοιχεία της λίστας έχουν παραλειφθεί - στο παράδειγμά μας τα περιεχόμενα του σακιδίου είναι μια υπακολουθία του ντουλαπιού

Περισσότερη Prolog

54

Κάποια παραδείγματα

```
?- subseq([1,3],[1,2,3,4]).
Yes
?- subseq(X,[1,2,3]).
X = [1, 2, 3] ;
X = [1, 2] ;
X = [1, 3] ;
X = [1] ;
X = [2, 3] ;
X = [2] ;
X = [3] ;
X = [] ;
No
```

Παρατηρείστε ότι το κατηγορήμα **subseq/2** όχι μόνο μπορεί να ελέγξει κατά πόσο μια λίστα είναι μια υπακολουθία κάποιας άλλης αλλά μπορεί επίσης να παράξει υπακολουθίες, τις οποίες και θα χρησιμοποιήσουμε για τη λύση του προβλήματος του σακιδίου.

Περισσότερη Prolog

55

```
/*
knapsackDecision(Items,Capacity,Goal,Knapsack)
takes a list Items of food terms, a positive number
Capacity, and a positive number Goal. We unify
Knapsack with a subsequence of Items representing
a knapsack with total calories >= Goal, subject to
the constraint that the total weight is <= Capacity.
*/
knapsackDecision(Items,Capacity,Goal,Knapsack) :-
subseq(Knapsack,Items),
weight(Knapsack,Weight),
Weight <= Capacity,
calories(Knapsack,Calories),
Calories >= Goal.
```

Περισσότερη Prolog

56

Για να δούμε που βρισκόμαστε...

```
?- knapsackDecision(
| [food(bread,4,9200),
| food(pasta,2,4500),
| food(peanutButter,1,6700),
| food(babyFood,3,6900)],
| 4,
| 10000,
| X).
X = [food(pasta, 2, 4500), food(peanutButter, 1, 6700)]
Yes
```

- Η παραπάνω κλήση αποφασίζει εάν υπάρχει λύση που να ικανοποιεί το στόχο των ελάχιστων θερμίδων (10000)
- Όχι όμως υποχρεωτικά τη λύση που ζητάμε εμείς...

Περισσότερη Prolog

57

Αποφασισιμότητα και Βελτιστοποίηση

- Λύσαμε το πρόβλημα της απόφασης για το πρόβλημα του σακιδίου
- Αυτό που θέλουμε να λύσουμε είναι το πρόβλημα της βελτιστοποίησης της λύσης του συγκεκριμένου προβλήματος
- Για να το επιτύχουμε, θα χρησιμοποιήσουμε ένα άλλο προκαθορισμένο κατηγορήμα της Prolog: `findall/3`

Περισσότερη Prolog

58

Το κατηγορήμα findall/3

- `findall(X, Goal, L)`
 - Βρίσκει όλους τους τρόπους με τους οποίους ο στόχος `Goal` ικανοποιείται
 - Για τον καθένα από αυτούς, εφαρμόζει στον όρο `X` την ίδια αντικατάσταση με την οποία ο στόχος `Goal` ικανοποιήθηκε
 - Ενοποιεί τη λίστα `L` με τη λίστα όλων αυτών των `X`

```
?- findall(a, subseq(_, [1,2]), L).
L = [a, a, a, a]
Yes
```

- Υπάρχουν τέσσερις λύσεις για το `subseq(_, [1, 2])`
- Συλλέξαμε μια λίστα από `a`, ένα για κάθε τρόπο απόδειξης

Περισσότερη Prolog

59

Συλλέγοντας τις απαντήσεις που θέλουμε

```
?- findall(X, subseq(X, [1,2]), L).
X = _G396
L = [[1, 2], [1], [2], []]
Yes
```

- Συνήθως, η πρώτη παράμετρος του `findall/3` είναι οι μεταβλητές που υπάρχουν στο στόχο προς επίλυση
- Η παραπάνω ερώτηση συλλέγει όλες τις απαντήσεις της ερώτησης-στόχου `subseq(X, [1, 2])`

Περισσότερη Prolog

60

```

/*
legalKnapsack(Items,Capacity,Knapsack) takes a list
Items of food terms and a positive number Capacity.
We unify Knapsack with a subsequence of Items whose
total weight is =< Capacity.
*/
legalKnapsack(Items,Capacity,Knapsack):-
subseq(Knapsack,Items),
weight(Knapsack,W),
W =< Capacity.

```

Περισσότερη Prolog

61

```

/*
maxCalories(List,Result) takes a List of lists of
food terms. We unify Result with an element from the
list that maximizes the total calories. We use a
helper predicate maxC that takes four parameters: the
remaining list of lists of food terms, the best list
of food terms seen so far, its total calories, and
the final result.
*/
maxCalories([First | Rest],Result) :-
calories(First,FirstC),
maxC(Rest,First,FirstC,Result).
maxC([],Sofar,_,Sofar).
maxC([First | Rest],_,MC,Result) :-
calories(First,FirstC),
MC =< FirstC,
maxC(Rest,First,FirstC,Result).
maxC([First | Rest],Sofar,MC,Result) :-
calories(First,FirstC),
MC > FirstC,
maxC(Rest,Sofar,MC,Result).

```

Περισσότερη Prolog

62

```

/*
knapsackOptimization(Items,Capacity,Knapsack) takes
a list Items of food items and a positive integer
Capacity. We unify Knapsack with a subsequence of
Items representing a knapsack of maximum total
calories, subject to the constraint that the total
weight is =< Capacity.
*/
knapsackOptimization(Items, Capacity, Knapsack) :-
findall(K, legalKnapsack(Items,Capacity,K), L),
maxCalories(L, Knapsack).

```

Περισσότερη Prolog

63

```

?- knapsackOptimization(
| [food(bread,4,9200),
| food(pasta,2,4500),
| food(peanutButter,1,6700),
| food(babyFood,3,6900)],
| 4,
| Knapsack).

Knapsack = [food(peanutButter, 1, 6700),
            food(babyFood, 3, 6900)]

Yes


```

Περισσότερη Prolog

64

Το πρόβλημα των οκτώ βασίλισσών (8-queens)

- Κάποιες πληροφορίες για το σκάκι:
 - Παίζεται σ' ένα τετράγωνο 8×8
 - Μια βασίλισσα μπορεί να κινηθεί οριζόντια, κάθετα ή διαγώνια για όσα τετράγωνα θελήσει
 - Δύο βασίλισσες απειλούν η μία την άλλη εάν είναι στην ίδια γραμμή, στήλη ή διαγώνιο (και η μία μπορεί να κινηθεί άμεσα στο τετράγωνο της άλλης)
- Το πρόβλημα: βρες ένα τρόπο να τοποθετηθούν οκτώ βασίλισσες σε μια κενή σκακιέρα έτσι ώστε καμία βασίλισσα να μην απειλείται από κάποια άλλη



Περισσότερη Prolog

65

Αναπαράσταση

- Θα μπορούσαμε να αναπαραστήσουμε τη βασίλισσα στη στήλη 2, σειρά 5 με τον όρο `queen(2,5)`
- Αλλά αφού δεν υπάρχουν άλλα κομμάτια στη σκακιέρα—π.χ. δεν θα έχουμε `pawn(X,Y)` ή `king(X,Y)`—θα χρησιμοποιήσουμε απλώς έναν όρο της μορφής `X/Y`
- (Δε θα τον αποτιμήσουμε ως διαίρεση)

Περισσότερη Prolog

66

Παράδειγμα

- Ένας σχηματισμός σκακιέρας είναι μια λίστα από βασίλισσες
- Ο παρακάτω είναι για τις βασίλισσες [2/5, 3/7, 6/1]

8								
7			Q					
6								
5	Q							
4								
3								
2								
1						Q		
	1	2	3	4	5	6	7	8

Περισσότερη Prolog

67

```

/*
nocheck(X/Y,L) takes a queen X/Y and a list
of queens. It succeeds if and only if the X/Y
queen holds none of the others in check.
*/
nocheck(_, []).
nocheck(X/Y, [X1/Y1 | Rest]) :-
    X =\= X1,
    Y =\= Y1,
    abs(Y1-Y) =\= abs(X1-X),
    nocheck(X/Y, Rest).

/*
legal(L) succeeds if L is a legal placement of
queens: all coordinates in range and no queen
in check.
*/
legal([]).
legal([X/Y | Rest]) :-
    legal(Rest),
    member(X, [1,2,3,4,5,6,7,8]),
    member(Y, [1,2,3,4,5,6,7,8]),
    nocheck(X/Y, Rest).

```

68

Επάρκεια

- Έχουμε ήδη αρκετά συστατικά για να λύσουμε το πρόβλημα: η ερώτηση `legal(X)` βρίσκει όλους τους επιτρεπούς σχηματισμούς:

```

?- legal(X).
X = [] ;
X = [1/1] ;
X = [1/2] ;
X = [1/3]

```

Περισσότερη Prolog

69

Λύση για τις οκτώ βασίλισσες

- Φυσικά, η παραπάνω «λύση» θα πάρει πολύ χρόνο: θα αρχίσει με το να βρει όλες τις 64 λύσεις με μια βασίλισσα, και μετά θα αρχίσει με όλες τις λύσεις με δύο, κοκ.
- Μπορούμε να «εκβιάσουμε» μια λύση με 8 βασίλισσες με την ερώτηση:

```

?- X = [_/_/_/_/_/_/_/_],
   | legal(X).
X = [8/4, 7/2, 6/7, 5/3,
     4/6, 3/8, 2/5, 1/1]
Yes

```

8			Q					
7					Q			
6				Q				
5	Q							
4							Q	
3				Q				
2							Q	
1	Q							
	1	2	3	4	5	6	7	8

Περισσότερη Prolog

70

Υπάρχει δυνατότητα για βελτίωση

- Η εύρεση της λύσης παίρνει πολύ χρόνο
- Στη συνέχεια, η Prolog βρίσκει απλές αναδιατάξεις της πρώτης λύσης:

```

?- X = [_/_/_/_/_/_/_/_], legal(X).
X = [8/4, 7/2, 6/7, 5/3, 4/6, 3/8, 2/5, 1/1] ;
X = [7/2, 8/4, 6/7, 5/3, 4/6, 3/8, 2/5, 1/1] ;
X = [8/4, 6/7, 7/2, 5/3, 4/6, 3/8, 2/5, 1/1] ;
X = [6/7, 8/4, 7/2, 5/3, 4/6, 3/8, 2/5, 1/1]

```

Περισσότερη Prolog

71

Βελτίωση του προγράμματος

- Προφανώς κάθε λύση έχει μία βασίλισσα σε κάθε στήλη
- Άρα, κάθε λύση μπορεί να γραφεί με τη μορφή:

$$X = [1/_/, 2/_/, 3/_/, 4/_/, 5/_/, 6/_/, 7/_/, 8/_/]$$
- Δίνοντας έναν όρο αυτής της μορφής περιορίζουμε την αναζήτηση (επιταχύνοντάς την) και αποφεύγουμε την εύρεση απλών αναδιατάξεων

```

/*
eightqueens(X) succeeds if X is a legal
placement of eight queens, listed in order
of their X coordinates.
*/
eightqueens(X) :-
    X = [1/_/, 2/_/, 3/_/, 4/_/, 5/_/, 6/_/, 7/_/, 8/_/],
    legal(X).

```

Περισσότερη Prolog

72

Περισσότερες βελτιώσεις

- Επειδή ξέρουμε ότι όλες οι Χ-συντεταγμένες είναι εντός του διαστήματος και διαφορετικές μεταξύ τους, το πρόγραμμα μπορεί λιγάκι να βελτιωθεί

```
nocheck(_, []).
nocheck(X/Y, [X1/Y1 | Rest]) :-
  % X =\= X1, assume the X's are distinct
  Y =\= Y1,
  abs(Y1-Y) =\= abs(X1-X),
  nocheck(X/Y, Rest).

legal([]).
legal([X/Y | Rest]) :-
  legal(Rest),
  % member(X, [1,2,3,4,5,6,7,8]), assume X in range
  member(Y, [1,2,3,4,5,6,7,8]),
  nocheck(X/Y, Rest).
```

Περισσότερη Prolog

73

Βελτιωμένη λύση στο πρόβλημα

- Το πρόγραμμα τώρα τρέχει αρκετά πιο γρήγορα
- Για παράδειγμα, δε βρίσκει πλέον όλες τις αναδιατάξεις

```
?- eightqueens(X).
X = [1/4, 2/2, 3/7, 4/3, 5/6, 6/8, 7/5, 8/1] ;
X = [1/5, 2/2, 3/4, 4/7, 5/3, 6/8, 7/6, 8/1]
```

Περισσότερη Prolog

74

Ένα πείραμα

```
legal([]).
legal([X/Y | Rest]) :-
  legal(Rest),
  % member(X, [1,2,3,4,5,6,7,8]), assume X in range
  1 =< Y, Y =< 8, % was member(Y, [1,2,3,4,5,6,7,8]),
  nocheck(X/Y, Rest).
```

- Το παραπάνω κατηγορήμα δε δουλεύει
- Εγείρει εξαίρεση: "arguments not sufficiently instantiated"
- Η κλήση του κατηγορήματος `member/2` δεν είναι απλώς κάποιος έλεγχος ότι οι συντεταγμένες είναι εντός ορίων αλλά είναι παραγωγή των συντεταγμένων

Περισσότερη Prolog

75

Άλλο ένα πείραμα

```
legal([]).
legal([X/Y | Rest]) :-
  % member(X, [1,2,3,4,5,6,7,8]), assume X in range
  member(Y, [1,2,3,4,5,6,7,8]),
  nocheck(X/Y, Rest),
  legal(Rest). % formerly the first condition
```

- Εγείρει εξαίρεση: "arguments not sufficiently instantiated"
- Η κλήση `legal(Rest)` πρέπει να προηγείται διότι παράγει την μερική λύση που χρειάζεται η αναδρομική κλήση του `nocheck`

Περισσότερη Prolog

76

Εύρεση μίας μόνο λύσης

- Αν θέλουμε να βρούμε μία μόνο λύση του προβλήματος μπορούμε να χρησιμοποιήσουμε οποιοδήποτε από τους δύο παρακάτω ορισμούς του σχετικού κατηγορήματος

```
/*
  eightqueens1(X) finds
  one solution to the
  eight queens problem.
*/
eightqueens1(X) :-
  once(eightqueens(X)).

/*
  eightqueens1(X) finds
  one solution to the
  eight queens problem.
*/
eightqueens1(X) :-
  eightqueens(X),
  !.
```

Περισσότερη Prolog

77

Το cut (!) της Prolog

- Το προκαθορισμένο κατηγορήμα `!/0` (διαβάζεται cut) χρησιμοποιείται για να περιορίσει την οπισθοδρόμηση
- Η εκτέλεσή του πάντα επιτυγχάνει και «κλαδεύει»:
 - Όλους τους εναλλακτικούς τρόπους ικανοποίησης των υποστόχων που πιθανώς να υπάρχουν μεταξύ της κεφαλής ενός κανόνα και του σημείου που το `!` εμφανίζεται στον κανόνα, και
 - Όλους τους κανόνες που μπορεί να έπονται του κανόνα που περιέχει το `!`
- Η χρήση του πρέπει να γίνεται με φειδώ και ύστερα από αρκετή σκέψη

Περισσότερη Prolog

78

Παράδειγμα: Οπισθοδρόμηση χωρίς και με cut

<code>p(X,Y) :- x(X), y(Y).</code>	<code>?- p(X,Y). X = 1 Y = a ;</code>	<code>p(X,Y) :- x(X), !, y(Y).</code>	<code>?- p(X,Y). X = 1 Y = a ;</code>
<code>p(X,Y) :- z(X,Y).</code>	<code>X = 1 Y = b ;</code>	<code>p(X,Y) :- z(X,Y).</code>	<code>X = 1 Y = b ;</code>
<code>p(4,d).</code>	<code>X = 2 Y = a ;</code>	<code>p(4,d).</code>	<code>No</code>
<code>x(1).</code>	<code>X = 2</code>	<code>x(1).</code>	
<code>x(2).</code>	<code>X = 2</code>	<code>x(2).</code>	
<code>y(a).</code>	<code>X = 2</code>	<code>y(a).</code>	
<code>y(b).</code>	<code>Y = b ;</code>	<code>y(b).</code>	
<code>z(3,c).</code>	<code>X = 3 Y = c ;</code>	<code>z(3,c).</code>	
	<code>X = 4 Y = d ;</code>		
	<code>No</code>		

Περισσότερη Prolog

79

Μέρη της Prolog που δεν εξετάσαμε

- Τη δυνατότητα ορισμού νέων τελεστών
- Το χειρισμό εξαιρέσεων
 - Εξαιρέσεις που εγείρονται από το σύστημα και από το χρήστη
 - Τα σχετικά κατηγορήματα `throw` και `catch`
- Τις βιβλιοθήκες και πολλά από τα προκαθορισμένα κατηγορήματα της γλώσσας

Περισσότερη Prolog

80

Αποχαιρετισμός στην Prolog

- Δε χρειάστηκε να παραλείψουμε τόσο ποσοστό της Prolog όσο παραλείψαμε στην ML και στη Java
- Η Prolog είναι μια μικρή γλώσσα
- Όμως είναι αρκετά ισχυρή ως εργαλείο επίλυσης (κάποιου είδους) προβλημάτων και ο επιδέξιος χειρισμός της από τον προγραμματιστή δεν είναι πολύ εύκολος
- Προϋποθέτει την υιοθέτηση μιας διαφορετικής φιλοσοφίας προγραμματισμού

Περισσότερη Prolog

81